// File: src/TimelockVesting.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.19;

import "@openzeppelin/token/ERC20/utils/SafeERC20.sol";
import "./TwoStepOwnable.sol";

/**
 * @title TimelockVesting
 * @notice A token holder contract that can release its token balance gradually like a
 *         typical vesting scheme, with a cliff and vesting period. Owner has the power
 *         to change the beneficiary who receives the vested tokens.
 * @author Hourglass Foundation
 */
contract TimelockVesting is TwoStepOwnable {
    using SafeERC20 for IERC20;

    error InvalidTotalAmount();
    error InvalidAmount();
    error InvalidBeneficiary();
    error InvalidStartTimestamp();
    error InvalidCliffStart(uint256 cliffDate, uint256 paramTimestamp);
    error InvalidDuration();
    error InvalidReleaseAmount();

    /// @notice The vesting token
    address internal _vestingToken;

    /// @notice Vesting schedule parameters
    /// @param amount amount of tokens to be vested
    /// @param startTimestamp unix timestamp of the start of vesting
    /// @param cliff unix timestamp of the cliff, before which no vesting counts
    /// @param duration duration in seconds of the period in which the tokens will vest
    /// @param released amount of tokens released
    struct VestingSchedule {
        uint256 amount;
        uint256 startTimestamp;
        uint256 cliff;
        uint256 duration;
        uint256 released;
    }

    /// @notice Internal storage of beneficiary address --> vesting schedule array
    mapping(address => VestingSchedule[]) internal _schedules;


    /// @param __vestingToken address of the token that is subject to vesting
    constructor(address __vestingToken) {
        _setInitialOwner(msg.sender);
        require(__vestingToken != address(0));
        _vestingToken = __vestingToken;
    }

    /// @notice Allows for adding multiple vests with a single call.
    /// @param _beneficiaries array of addresses of the beneficiaries to whom vested tokens are transferred
    /// @param _amounts array of amounts of tokens to be vested
    /// @param _startTimestamps array of unix timestamps of the start of vesting
    /// @param _cliffDates array of unix timestamps of the cliff, before which no vesting counts
    /// @param _durationInSeconds array of durations in seconds of the period in which the tokens will vest
    /// @param _totalBatchTokenAmount total amount of tokens to be vested across all beneficiaries (sanity check)
    /// @dev Call this with a script that validates that all arrays are equal length & that
    ///      all addresses, amounts, timestamps & durations != 0
    function addBatchBeneficiaries(
        address[] calldata _beneficiaries,
        uint256[] calldata _amounts,
        uint256[] calldata _startTimestamps,
        uint256[] calldata _cliffDates,
        uint256[] calldata _durationInSeconds,
        uint256 _totalBatchTokenAmount
    ) external onlyOwner {
        // pull in the tokens for this batch
        IERC20(_vestingToken).safeTransferFrom(msg.sender, address(this), _totalBatchTokenAmount);

        uint256 totalVestingAmount;

        // add each vesting schedule
        uint256 numVests = _beneficiaries.length;
        for (uint256 i; i < numVests;) {
            _addBeneficiary(
                _beneficiaries[i],
                _amounts[i],
                _startTimestamps[i],
                _cliffDates[i],
                _durationInSeconds[i]
            );

            totalVestingAmount += _amounts[i];

            unchecked {
                ++i;
            }
        }
        
        if (totalVestingAmount != _totalBatchTokenAmount) {
            revert InvalidTotalAmount();
        }
    }

    /// @notice Initializes a vesting contract that vests its balance of any ERC20 token to the
    ///    _beneficiary in a linear fashion until duration has passed. By then all
    ///    of the balance will have vested.
    /// @param _beneficiary address of the beneficiary to whom vested tokens are transferred
    /// @param _amount amount of tokens to be vested
    /// @param _startTimestamp unix timestamp of the start of vesting
    /// @param _cliffDate unix timestamp of the cliff, before which no vesting counts
    /// @param _durationInSeconds duration in seconds of the period in which the tokens will vest
    function addBeneficiary(
        address _beneficiary,
        uint256 _amount,
        uint256 _startTimestamp,
        uint256 _cliffDate,
        uint256 _durationInSeconds
    ) public onlyOwner {
        // pull in the tokens for this beneficiary
        IERC20(_vestingToken).safeTransferFrom(msg.sender, address(this), _amount);

        _addBeneficiary(_beneficiary, _amount, _startTimestamp, _cliffDate, _durationInSeconds);
    }

    function _addBeneficiary(
        address _beneficiary,
        uint256 _amount,
        uint256 _startTimestamp,
        uint256 _cliffDate,
        uint256 _durationInSeconds
    ) internal {
        // sanity checks
        if (_amount == 0) revert InvalidAmount();
        if (_beneficiary == address(0)) revert InvalidBeneficiary();
        if (_startTimestamp == 0) revert InvalidStartTimestamp();
        if (_durationInSeconds == 0) revert InvalidDuration();
        if (_cliffDate <= _startTimestamp) revert InvalidCliffStart(_cliffDate, _startTimestamp);
        if (_cliffDate > _startTimestamp + _durationInSeconds) revert InvalidCliffStart(_cliffDate, _startTimestamp + _durationInSeconds);

        // add the vesting schedule
        _schedules[_beneficiary].push(VestingSchedule(
            _amount,
            _startTimestamp,
            _cliffDate,
            _durationInSeconds,
            0
        ));

        emit VestingInitialized(
            _beneficiary,
            _amount,
            _startTimestamp,
            _cliffDate,
            _durationInSeconds
        );
    }

    /// @notice Transfers vested tokens to beneficiary.
    /// @dev This won't revert if nothing is claimable, but nothing will be claimed.
    /// @param _beneficiary Address of beneficiary to claim tokens for.
    /// @param _scheduleId Id of schedule to claim tokens for.
    function release(address _beneficiary, uint256 _scheduleId) external {
        if (_beneficiary == address(0)) revert InvalidBeneficiary();

        // Transfer vested tokens to beneficiary
        IERC20(_vestingToken).safeTransfer(_beneficiary, _release(_beneficiary, _scheduleId));//releasedAmt);
    }

    /// @notice Allows beneficiary to claim vested tokens across multiple schedules.
    /// @dev This won't revert if nothing is claimable, but nothing will be transferred.
    /// @param _beneficiary Address of beneficiary to claim tokens for.
    /// @param _scheduleIds Array of schedule ids to claim tokens for.
    function releaseMultiple(address _beneficiary, uint256[] calldata _scheduleIds) external {
        uint256 numSchedules = _scheduleIds.length;
        uint256 releasedAmt;
        for (uint256 i; i < numSchedules; i++) {
            releasedAmt += _release(_beneficiary, _scheduleIds[i]);
        }

        // transfer vested tokens to beneficiary
        IERC20(_vestingToken).safeTransfer(_beneficiary, releasedAmt);
    }

    /// @notice Updates the amount released & returns the amount to distribute for this `_scheduleId`.
    function _release(address _beneficiary, uint256 _scheduleId) internal returns (uint256 vested) {
        VestingSchedule memory schedule = _schedules[_beneficiary][_scheduleId];

        vested = _vestedAmount(schedule);

        // instead of reverting, return 0 if nothing is due so multiple schedules can be checked
        if (vested == 0) {
            return 0;
        }

        // otherwise updated released amount
        _schedules[_beneficiary][_scheduleId].released += vested;

        // sanity check
        if (schedule.released > schedule.amount) revert InvalidReleaseAmount();

        emit Released(_beneficiary, _scheduleId, vested);
    }

    /// @notice Calculates the amount that has already vested but hasn't been released yet.
    function _vestedAmount(VestingSchedule memory schedule) internal view returns (uint256) {
        // cliff hasn't passed yet neither has start time, so 0 vested
        if (block.timestamp < schedule.cliff) {
            return 0;
        }

        uint256 elapsedTime = block.timestamp - schedule.startTimestamp;

        // If over vesting duration, all tokens vested
        if (elapsedTime >= schedule.duration) {
            // deduct already released tokens
            return schedule.amount - schedule.released; 
        } else {
            // if 75 seconds have passed of the 100 seconds, then 3/4 of the amount should be released.
            uint256 vested = schedule.amount * elapsedTime / schedule.duration;
            return (vested - schedule.released); 
        }
    }

    /// @notice Changes beneficiary who receives the vested token.
    /// @dev Only governance can call this function. This is to be used in case the target address
    ///   needs to be updated. If the previous beneficiary has any unclaimed tokens, the new beneficiary
    ///   will be able to claim them and the rest of the vested tokens.
    /// @param oldBeneficiary address of the previous beneficiary
    /// @param newBeneficiary new address to become the beneficiary
    /// @param scheduleIds array of schedule ids to migrate to the new beneficiary
    function changeBeneficiary(address oldBeneficiary, address newBeneficiary, uint256[] calldata scheduleIds) external onlyOwner {
        if (newBeneficiary == address(0)) revert InvalidBeneficiary();
        if (newBeneficiary == oldBeneficiary) revert InvalidBeneficiary();

        uint256 numSchedules = scheduleIds.length;
        // iterate from the end to avoid having to delete & move every schedule element each iteration
        for (uint256 i; i < numSchedules; i++) {
            VestingSchedule memory schedule = _schedules[oldBeneficiary][scheduleIds[i]];

            // migrate the schedule to the new beneficiary
            _schedules[newBeneficiary].push(schedule);

            // rather than deleting, set amount to the amount released  & duration to 0
            //   to avoid having to delete & move every schedule element each iteration
            //   and to avoid breaking the amount released logic in _vestedAmount()
            _schedules[oldBeneficiary][scheduleIds[i]].amount = schedule.released;
            _schedules[oldBeneficiary][scheduleIds[i]].duration = 0;
            emit SetBeneficiary(oldBeneficiary, scheduleIds[i], newBeneficiary, _schedules[newBeneficiary].length - 1);
        }
    }

    /// @notice Allows the owner to terminate a vesting schedule and recover the unvested tokens.
    /// @param _beneficiary Address of beneficiary to terminate vesting for.
    /// @param _scheduleId Id of schedule to terminate.
    function terminateVest(address _beneficiary, uint256 _scheduleId) external onlyOwner {
        // _release(_beneficiary, _scheduleId);

        VestingSchedule memory schedule = _schedules[_beneficiary][_scheduleId];
        // calculate the amount accrued to date
        uint256 vested = _vestedAmount(schedule);// - schedule.released;
        uint256 unreleased = schedule.amount - (schedule.released + vested);

        // delete the vest
        delete _schedules[_beneficiary][_scheduleId];

        // transfer out the tokens
        IERC20(_vestingToken).safeTransfer(msg.sender, unreleased);
        IERC20(_vestingToken).safeTransfer(_beneficiary, vested);
    }


    ////////// Getter Functions //////////

    /// @notice Checks the amount of currently vested tokens available for release.
    /// @dev Note that this will return a value > 0 if there are any tokens available to claim,
    /// @param _beneficiary The address of the beneficiary.
    /// @param _scheduleId The index of the schedule array.
    /// @return The number of tokens that are vested and available to claim.
    function getClaimableAmount(address _beneficiary, uint256 _scheduleId) external view returns (uint256) {
        VestingSchedule memory schedule = _schedules[_beneficiary][_scheduleId];
        return _vestedAmount(schedule);
    }

    /// @notice Obtain a specific schedule for a user.
    /// @dev Note that this will return a schedule even if it has been transferred.
    /// @param _beneficiary The address of the beneficiary.
    /// @param _scheduleId The index of the schedule array.
    /// @return The vesting schedule.
    function getSchedule(address _beneficiary, uint256 _scheduleId) external view returns (VestingSchedule memory) {
        return _schedules[_beneficiary][_scheduleId];
    }

    /// @notice Obtain the length of a user's schedule array.
    /// @dev Note that this will return a length that includes deleted schedules.
    /// @param _beneficiary The address of the beneficiary.
    /// @return The length of the schedule array.
    function getNumberSchedules(address _beneficiary) external view returns (uint256) {
        return _schedules[_beneficiary].length;
    }

    /// @notice Obtain the total amount of tokens released to a user thus far.
    /// @param _beneficiary The address of the beneficiary.
    /// @return The total amount of tokens released.
    function getTotalAmountReleased(address _beneficiary) external view returns (uint256) {
        uint256 numSchedules = _schedules[_beneficiary].length;
        uint256 totalReleased;
        for (uint256 i; i < numSchedules; i++) {
            totalReleased += _schedules[_beneficiary][i].released;
        }
        return totalReleased;
    }

    /// @notice Get the token being vested.
    /// @return The token address.
    function vestingToken() public view returns (address) {
        return _vestingToken;
    }

    
    ////////// EVENTS //////////
    /// @notice Emitted when a beneficiary claims vested tokens.
    event Released(address indexed beneficiary, uint256 scheduleId, uint256 amount);
    /// @notice Emitted when a new vesting schedule is created.
    event VestingInitialized(
        address indexed beneficiary, 
        uint256 amount,
        uint256 startTimestamp,
        uint256 cliff,
        uint256 duration
    );
    /// @notice Emitted when a beneficiary is changed.
    event SetBeneficiary(address indexed oldBeneficiary, uint256 oldBeneficiaryScheduleIndex, address indexed newBeneficiary, uint256 newBeneficiaryScheduleIndex);
    /// @notice Emitted when whether votes can be counted is changed.
    event BeneficiaryVotes(bool voting);
}
// File: lib/openzeppelin-contracts/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: src/TwoStepOwnable.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.18;

import "./TwoStepOwnableInterface.sol";

/**
 * @title   TwoStepOwnable
 * @author  OpenSea Protocol Team
 * @notice  TwoStepOwnable provides access control for inheriting contracts,
 *          where the ownership of the contract can be exchanged via a two step
 *          process. A potential owner is set by the current owner by calling
 *          `transferOwnership`, then accepted by the new potential owner by
 *          calling `acceptOwnership`.
 */
abstract contract TwoStepOwnable is TwoStepOwnableInterface {
    // The address of the owner.
    address private _owner;

    // The address of the new potential owner.
    address private _potentialOwner;

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        // Ensure that the caller is the owner.
        if (msg.sender != _owner) {
            revert CallerIsNotOwner();
        }

        // Continue with function execution.
        _;
    }

    /**
     * @notice Initiate ownership transfer by assigning a new potential owner
     *         to this contract. Once set, the new potential owner may call
     *         `acceptOwnership` to claim ownership. Only the owner may call
     *         this function.
     *
     * @param newPotentialOwner The address for which to initiate ownership
     *                          transfer to.
     */
    function transferOwnership(address newPotentialOwner)
        external
        override
        onlyOwner
    {
        // Ensure the new potential owner is not an invalid address.
        if (newPotentialOwner == address(0)) {
            revert NewPotentialOwnerIsNullAddress();
        }

        // Emit an event indicating that the potential owner has been updated.
        emit PotentialOwnerUpdated(newPotentialOwner);

        // Set the new potential owner as the potential owner.
        _potentialOwner = newPotentialOwner;
    }

    /**
     * @notice Clear the currently set potential owner, if any. Only the owner
     *         of this contract may call this function.
     */
    function cancelOwnershipTransfer() external override onlyOwner {
        // Emit an event indicating that the potential owner has been cleared.
        emit PotentialOwnerUpdated(address(0));

        // Clear the current new potential owner.
        delete _potentialOwner;
    }

    /**
     * @notice Accept ownership of this contract. Only the account that the
     *         current owner has set as the new potential owner may call this
     *         function.
     */
    function acceptOwnership() external override {
        // Ensure the caller is the potential owner.
        if (msg.sender != _potentialOwner) {
            // Revert, indicating that caller is not current potential owner.
            revert CallerIsNotNewPotentialOwner();
        }

        // Emit an event indicating that the potential owner has been cleared.
        emit PotentialOwnerUpdated(address(0));

        // Clear the current new potential owner.
        delete _potentialOwner;

        // Set the caller as the owner of this contract.
        _setOwner(msg.sender);
    }

    /**
     * @notice An external view function that returns the potential owner.
     *
     * @return The address of the potential owner.
     */
    function potentialOwner() external view override returns (address) {
        return _potentialOwner;
    }

    /**
     * @notice A public view function that returns the owner.
     *
     * @return The address of the owner.
     */
    function owner() public view override returns (address) {
        return _owner;
    }

    /**
     * @notice Internal function that sets the inital owner of the base
     *         contract. The initial owner must not already be set.
     *         To be called in the constructor or when initializing a proxy.
     *
     * @param initialOwner The address to set for initial ownership.
     */
    function _setInitialOwner(address initialOwner) internal {
        // Ensure that an initial owner has been supplied.
        if (initialOwner == address(0)) {
            revert InitialOwnerIsNullAddress();
        }

        // Ensure that the owner has not already been set.
        if (_owner != address(0)) {
            revert OwnerAlreadySet(_owner);
        }

        // Set the initial owner.
        _setOwner(initialOwner);
    }

    /**
     * @notice Private function that sets a new owner and emits a corresponding
     *         event.
     *
     * @param newOwner The address to assign as the new owner.
     */
    function _setOwner(address newOwner) private {
        // Emit an event indicating that the new owner has been set.
        emit OwnershipTransferred(_owner, newOwner);

        // Set the new owner.
        _owner = newOwner;
    }
}
// File: lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: lib/openzeppelin-contracts/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: lib/openzeppelin-contracts/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: src/TwoStepOwnableInterface.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.18;

/**
 * @title   TwoStepOwnableInterface
 * @author  OpenSea Protocol
 * @notice  TwoStepOwnableInterface contains all external function interfaces,
 *          events and errors for the TwoStepOwnable contract.
 */

interface TwoStepOwnableInterface {
    /**
     * @dev Emit an event whenever the contract owner registers a new potential
     *      owner.
     *
     * @param newPotentialOwner The new potential owner of the contract.
     */
    event PotentialOwnerUpdated(address newPotentialOwner);

    /**
     * @dev Emit an event whenever contract ownership is transferred.
     *
     * @param previousOwner The previous owner of the contract.
     * @param newOwner      The new owner of the contract.
     */
    event OwnershipTransferred(address previousOwner, address newOwner);

    /**
     * @dev Revert with an error when attempting to set an initial owner when
     *      one has already been set.
     */
    error OwnerAlreadySet(address currentOwner);

    /**
     * @dev Revert with an error when attempting to call a function with the
     *      onlyOwner modifier from an account other than that of the owner.
     */
    error CallerIsNotOwner();

    /**
     * @dev Revert with an error when attempting to register an initial owner
     *      and supplying the null address.
     */
    error InitialOwnerIsNullAddress();

    /**
     * @dev Revert with an error when attempting to register a new potential
     *      owner and supplying the null address.
     */
    error NewPotentialOwnerIsNullAddress();

    /**
     * @dev Revert with an error when attempting to claim ownership of the
     *      contract with a caller that is not the current potential owner.
     */
    error CallerIsNotNewPotentialOwner();

    /**
     * @notice Initiate ownership transfer by assigning a new potential owner
     *         to this contract. Once set, the new potential owner may call
     *         `acceptOwnership` to claim ownership. Only the owner may call
     *         this function.
     *
     * @param newPotentialOwner The address for which to initiate ownership
     *                          transfer to.
     */
    function transferOwnership(address newPotentialOwner) external;

    /**
     * @notice Clear the currently set potential owner, if any. Only the owner
     *         of this contract may call this function.
     */
    function cancelOwnershipTransfer() external;

    /**
     * @notice Accept ownership of this contract. Only the account that the
     *         current owner has set as the new potential owner may call this
     *         function.
     */
    function acceptOwnership() external;

    /**
     * @notice An external view function that returns the potential owner.
     *
     * @return The address of the potential owner.
     */
    function potentialOwner() external view returns (address);

    /**
     * @notice An external view function that returns the owner.
     *
     * @return The address of the owner.
     */
    function owner() external view returns (address);
}